Modelling Radioactive Decay Using Google Sheets

This activity uses the random number generator in Google Sheets to model radioactive decay. Each unstable nucleus is represented by a cell on the spreadsheet, and the probability that it decays each unit of time can be varied.

Accessing Google Sheets

If you already have a Google or Gmail account, you will already be able to access Google Sheets. Otherwise, you will need to set up a Google account. You can do this with any email address (you do not need to set up a new email address). Go to https://accounts.google.com/SignUpWithoutGmail to register and then verify your account by clicking on the link in your email.

Once you have an account, go to https://www.google.co.uk/sheets/about/ and select 'Go to Google Sheets'. You will then be prompted to log in, after which Google Sheets will open.

Your spreadsheet will be saved to your Google Drive. If you want to access it later, go to https://drive.google.com/drive/my-drive After logging in, you will be able to see any documents you have created.

Setting up your spreadsheet:

1. In cell A1, type 'Probability of decay, P'.

2. In cell B1, enter the value for the probability of decay. This can be anything between 0 and 1. For example, using a coin and 'Heads' to model decay would be a probability of 1/2. Rolling a dice and taking a '6' as indicating decay would be a probability of 1/6. Try 0.1 to start with.

3. In cell A3, type 'Time, t'.

4. In cells A4 to A104 enter the numbers 0 to 100. These are our units of time.

5. In cell B3, type 'Number of radioactive nuclei remaining, N'

   You will be plotting the time, t, against the number of radioactive nuclei remaining, N, to obtain your decay curve.

	A	B	C
1	**Probability of decay, P**	0.1	
2			
3	**Time, t**	**Number of radioactive nuclei remaining, N**	
4	0		
5	1		
6	2		
7	3		
8	4		
9	5		
10	6		
11	7		
12	8		
13	9		
14	10		
15	11		
16	12		
17	13		
18	14		
19	15		
20	16		
21	17		

Creating the nuclei:

6. Your nuclei will be in cells D4 to II4. This gives you 240 radioactive nuclei initially ($N_0 = 240$). However, you can have as many nuclei as you like! The more nuclei, the smoother your decay curve will be.

   We will use '1' to represent a nucleus that has not yet decayed and '0' to indicate a nucleus that has decayed. Therefore enter '1' in each box from D4 to II4 to show the state of all the nuclei at t=0. Note that Google sheets initially only gives you columns up to Z, but if you copy and paste a string of 1s into column Z it will automatically create new columns to accommodate these. Keep copying-and-pasting until you reach II.

7. In cell D5, we are going to calculate whether the nucleus represented by cell D4 has decayed yet.

   To generate a random number between 0 and 1 (which is how we will model the random nature of radioactive decay), we use '=RAND()' Our probability that a decay takes place is given in cell B1. We are therefore going to add an equation that states that '**If** (**IF**) the **random number** (**RAND()**) generated **is greater than** (**>**) our **decay probability** (**B1**), then the nucleus **remains radioactive** (**1**). Otherwise, **the nucleus decays** (**0**).' This is given by the equation '=IF(RAND()>$B$1,1,0)' Note that the $ signs mean that we always want to refer to cell B1.

   If the nucleus has already decayed at a previous time, we do not want it to 'un-decay' (change from a 0 to a 1). Therefore we multiply the output of the equation by the cell above. Therefore, in cell D5 you should enter =(IF(RAND()>=$B$1,1,0))*D4.

fx =IF(RAND()>$B$1,1,0)*D4

	A	B	C	D	E	F	G
1	Probablity of decay, P	0.1					
2							
3	Time, t	Number of radioactive nuclei remaining, N					
4	0			1	1	1	1
5	1			1			
6	2						
7	3						
8	4						
9	5						
10	6						
11	7						
12	8						
13	9						
14	10						
15	11						
16	12						

You will find that D5 becomes either a 1 or a 0.

8. Fill this equation across all 240 'nuclei' (ie cells D5 to II5). To do this, click in cell D5. Move your cursor to the small square in the bottom, right-hand corner. When your cursor turns into a small black cross, click and drag. With cells D5 to II5 still highlighted, use the small bottom right square again to fill down to row 104 (ie for all of the times).

ID	IE	IF	IG	IH	II	IJ
1	1	1	1	1	1	
1	1	0	1	1	1	

Click and drag from here

Taking your data:

9. We now need to fill in column B, the number of radioactive nuclei remaining. To do this we need to count how many 1s there are in each row. We are going to use the 'COUNTIF' function to do this.

   In cell B4, enter =COUNTIF(D4:II4,1) This will count all of the 1s from cell D4 to cell II4. You should get a value of 240.

10. Fill this equation down column B for all times (clicking and dragging, as in step 8).

fx =COUNTIF(D4:II4,1)

	A	B	C	D	E	F
1	Probablity of decay, P	0.1				
2						
3	Time, t	Number of radioactive nuclei remaining, N				
4	0	240		1	1	1
5	1	214		1	1	1
6	2	191		0	1	1
7	3	176		0	1	1
8	4	155		0	1	1
9	5	134		0	1	0
10	6	119		0	1	0
11	7	106		0	1	0
12	8	97		0	1	0
13	9	84		0	1	0
14	10	76		0	1	0
15	11	64		0	1	0
16	12	60		0	1	0
17	13	52		0	1	0
18	14	49		0	1	0
19	15	40		0	1	0

You are likely to find that your numbers differ slightly from this because of the random number generator.

11. We can now plot the graph to show the decay curve. Highlight your data in columns A and B and click Insert, Chart. This should automatically plot t on the x-axis and N on the y-axis.



Click 'Insert' from the top menu and then select 'Chart'

Your graph will appear the at the bottom of your data. It will initially be a line graph, but we can change this to a scatter graph using the options on the right of the screen.



Click on the down arrow at the side of 'Line chart' and from the options that appear, select 'Scatter'. (Note: If these options aren't visible, double click on your graph to pull up this side-bar).

12. Your graph should automatically have the axes labelled and a chart title. You can modify these using the 'Customise' tab of the right-hand 'Chart editor' menu.

    You can also change the scale of your graph by selecting 'Gridlines and ticks' and altering the 'Major step' for both the vertical and horizontal axes.



Interpreting your data:

13. Press the 'Backspace' or 'Delete' key in any empty cell of your spreadsheet to generate new random numbers and see how your graph changes slightly each time.

14. Estimate the half-life from your graph. You should take an average of several values (for example, how long does the number of radioactive nuclei take to fall from 240 to 120, from 100 to 50, from 60 to 30 etc)

15. Try changing the probability of decay (in cell B1) and see what happens to the shape of your graph. How does the affect the half-life?

Comparing your 'experimental data' to the theoretical values (optional):

16. In cell C3, type 'Theoretical Value, N'

17. In cell C4, we need to know our value of N_0. We are therefore going to use the 'Countif' function again: =COUNTIF(D4:II4,1). You should get a value of 240.

18. In order to generate theoretical values of N, we need to calculate the value of the decay constant (λ) based on the probability of decay. Starting from the equation for radioactive decay:

$$N(t) = N_0\, e^{-\lambda t}$$

$e^{-\lambda t}$ therefore gives us the probability that a nucleus remains radioactive. Therefore the probability of decay, P(t), is given by:

$$P(t) = 1 - e^{-\lambda t}$$

We now need to rearrange this to solve for the decay constant, λ.

$$e^{-\lambda t} = 1 - P(t)$$

Setting t=1 and taking the natural logarithm (ln) of both sides gives:

$$-\lambda = \ln(1 - P)$$

Therefore,

$$\lambda = -\ln(1 - P)$$

To enter this into the spreadsheet, type 'Decay constant, λ' into cell A2 and then our equation into cell B2 =-LN(1-$B$1)

19. To calculate the theoretical values of N, we will use the value of λ that we have just calculated, together with the decay equation:

$$N(t) = N_0\, e^{-\lambda t}$$

Therefore, in cell C5 enter =$C$4*EXP(-$B$2*A5)

20. Fill this equation down column C.

fx =$C$4*EXP(-$B$2*A5)

	A	B	C
1	Probablity of decay, P	0.1	
2	Decay constant, λ	0.1053605157	
3	Time, t	Number of radioactive nuclei remaining, N	Theoretical Value, N
4	0	240	240
5	1	214	216
6	2	192	194.4
7	3	176	174.96
8	4	159	157.464
9	5	141	141.7176
10	6	125	127.54584
11	7	114	114.791256
12	8	102	103.3121304
13	9	94	92.98091736
14	10	78	83.68282562
15	11	70	75.31454306
16	12	67	67.78308876
17	13	59	61.00477988
18	14	51	54.90430189
19	15	48	49.4138717
20	16	45	44.47248453
21	17	41	40.02523608
22	18	40	36.02271247

You should find that the theoretical values match exactly to those given above, whilst the 'experimental' values will vary.

21. As the value of N should be a whole number, highlight the cells. Select 'Format' from the top menu, go to 'Number', then 'More Formats', 'Custom number formats', select '0' and click 'Apply'.

22. You can now add the theoretical values of N onto you graph. In the 'Chart editor', select 'Add Series' under the 'Setup' tab.

Click on the grid that appears.

In the box that appears, enter your theoretical values C4:C104 and click 'OK'.





You should now see a second set of data points on your graph in a different colour.

23. By double clicking on the coloured dots at the top of your graph (under the title), you will be able to label the two colours to create a key.



24. Again, try pressing ‘Backspace’ or ‘Delete’ in an empty cell to generate new random numbers and see how this affects your decay curve in comparison to the theoretical values.

25. You can also try altering the probability of decay and see this reflected in both sets of values.